```
>
```

Integrals by Maple

```
> restart;
```

In this worksheet, we show how to integrate using Maple.

We consider the function $f(x) = \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}$. We enter this into Maple by using the Maple functional notation.

```
> f:=x -> (2*x^5-8*x^4+15*x^3-10*x^2-9*x+27)/(x^4-4*x^3+5*x^2-4*x+4);
```

$$f := x \to \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}$$

This allows for easy evaluation of the function for different values.

```
> f(3);
```

$$\frac{153}{10}$$

We indicate its integral by using the Int command. Int is an inert form of integration and is good for printing integrals, but does not evaluate.

```
> Int(f(x),x);
```

$$\int \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}\, dx$$

We use the value command to evaluate the integral. The % sign refers to the most recently computed item. Notice that the constant of integration is not included as part of the answer.

```
> value(%);
```

$$x^2 - \frac{5}{x-2} + 3\ln(x-2) + \ln(x^2+1) + 7\arctan(x)$$

The int command evaluates the indefinite integral by just producing its anti-derivative (without the constant of integration), and does not print the integral symbol.

```
> int(f(x),x);
```

$$x^2 - \frac{5}{x-2} + 3\ln(x-2) + \ln(x^2+1) + 7\arctan(x)$$

Int and int can also be used for definite integration as seen in the following. Suppose we wish to integrate our function from $x = -2$ to $x = 1$. We first plot this part of the graph to take a look at the function.

```
> plot(f(x),x=-2..1);
```



We notice that the function is continuous over our interval of integration.

```
> Int(f(x),x=-2..1);
```

$$\int_{-2}^{1} \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}\,dx$$

We evaluate this definite integral two different ways, by using first value, then int.

```
> % = value(%);
```

$$\int_{-2}^{1} \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}\,dx = -5\ln(2) + \frac{3}{4} + \frac{7}{4}\pi - \ln(5) + 7\arctan(2)$$

```
> %% = int(f(x),x=-2..1);
```

$$\int_{-2}^{1} \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}\,dx = -5\ln(2) + \frac{3}{4} + \frac{7}{4}\pi - \ln(5) + 7\arctan(2)$$

We can incorporate the evalf command to get a decimal approximation to the exact integral.

```
> evalf(int(f(x),x=-2..1));
```

$$8.922654355$$

Instead of entering a function first by using the Maple functional notation , we may also enter it as a Maple expression directly into the Int and int commands.

```
> Int((2*x^5-8*x^4+15*x^3-10*x^2-9*x+27)/(x^4-4*x^3+5*x^2-4*x+4),x)=int
  ((2*x^5-8*x^4+15*x^3-10*x^2-9*x+27)/(x^4-4*x^3+5*x^2-4*x+4),x);
```

$$\int \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}\,dx = x^2 - \frac{5}{x-2} + 3\ln(x-2) + \ln(x^2+1) + 7\arctan(x)$$

\>

We now look at several examples from class.

Example 1

```
> Int(exp(3*x)*cos(5*x),x)=int(exp(3*x)*cos(5*x),x);
```

$$\int \mathbf{e}^{(3x)} \cos(5x)\,dx = \frac{3}{34}\mathbf{e}^{(3x)}\cos(5x) + \frac{5}{34}\mathbf{e}^{(3x)}\sin(5x)$$

Example 2

```
> Int((x^2-3*x+1)*exp(5*x),x)=int((x^2-3*x+1)*exp(5*x),x);
```

$$\int (x^2 - 3x + 1)\,\mathbf{e}^{(5x)}\,dx = \frac{1}{5}\mathbf{e}^{(5x)}x^2 - \frac{17}{25}\mathbf{e}^{(5x)}x + \frac{42}{125}\mathbf{e}^{(5x)}$$

Example 3

```
> Int((sin(3*x))^4*(cos(3*x))^2,x)=int((sin(3*x))^4*(cos(3*x))^2,x);
```

$$\int \sin(3x)^4\cos(3x)^2\,dx =$$

$$-\frac{1}{18}\sin(3x)^3\cos(3x)^3 - \frac{1}{24}\sin(3x)\cos(3x)^3 + \frac{1}{48}\cos(3x)\sin(3x) + \frac{1}{16}x$$

This answer differs from the one we got in class, but we can check that they are the same by subtracting the right hand side (rhs) of the above from the answer obtained in class. We simultaneously simplify.

```
> simplify((1/18)*(sin(3*x))^5*cos(3*x)-(1/72)*(sin(3*x))^3*cos(3*x)-(1
  /48)*
  sin(3*x)*cos(3*x)+(1/16)*x-rhs(%));
```

$$0$$

Although we got 0 here, showing equality, it would be sufficient to get any real constant.

Example 4

```
> Int((cos(x))^4*(sin(x))^5,x)=int((cos(x))^4*(sin(x))^5,x);
```

$$\int \cos(x)^4\sin(x)^5\,dx = -\frac{1}{9}\sin(x)^4\cos(x)^5 - \frac{4}{63}\sin(x)^2\cos(x)^5 - \frac{8}{315}\cos(x)^5$$

Again, we compare this answer to the one obtained in class.

```
> simplify(-(cos(x))^9/9+2*(cos(x))^7/7-(cos(x))^5/5-rhs(%));
```

$$0$$

Example 5

```
> Int((4*x+3)/(x^2+7*x+10),x)=int((4*x+3)/(x^2+7*x+10),x);
```

$$\int \frac{4x+3}{x^2+7x+10}\,dx = \frac{17}{3}\ln(x+5) - \frac{5}{3}\ln(x+2)$$

Notice the lack of absolute values here. What is happening is that Maple assumes that **ln** is the complex valued version of the function.

Consider the function $f(x) = \frac{1}{x+5}$. This function is not continuous at -5. Let's look at its graph.

```
> f:=1/(x+5);
```

$$f := \frac{1}{x+5}$$

```
> plot(f,x=-10..5,y=-5..5,discont=true);
```



Now let us look at its indefinite integral.

```
> integral:=int(f,x);
```

$$integral := \ln(x+5)$$

Next let's see what happens for definite integrals, first using limits of integration to the right of the discontinuity.

```
> Int(f,x=-2..5)=int(f,x=-2..5);
```

$$\int_{-2}^{5} \frac{1}{x+5}\, dx = \ln(2) + \ln(5) - \ln(3)$$

Let's substitute these limits of integration into $\ln(|x+5|)$ as if we were using the Fundamental Theorem by hand.

```
> subs(x=5,ln(abs(x+5)))-subs(x=-2,ln(abs(x+5)));
```

$$\ln(|10|) - \ln(|3|)$$

Note that we get equal answers. Next let's try limits of integration to the left of the discontinuity.

```
> Int(f,x=-9..-6)=int(f,x=-9..-6);
```

$$\int_{-9}^{-6} \frac{1}{x+5}\, dx = -2\ln(2)$$

Again, let's substitute these limits of integration into $\ln(|x+5|)$.

```
> subs(x=-6,ln(abs(x+5)))-subs(x=-9,ln(abs(x+5)));
```

$$\ln(|-1|) - \ln(|-4|)$$

We again get anwers that are equal. Finally, lets try one limit of integration from each side of the discontinuity.

```
> Int(f,x=-10..4)=int(f,x=-10..4);
```

$$\int_{-10}^{4} \frac{1}{x+5}\,dx = \int_{-10}^{4} \frac{1}{x+5}\,dx$$

This indicates that Maple can't come up with an answer, which is good. so, even though Maple does not include the absolute value signs for **ln**, it does evaluate the definite integrals correctly.

Example 6

```
> Int((y^2-3)/(y^2+3),y)=int((y^2-3)/(y^2+3),y);
```

$$\int \frac{y^2-3}{y^2+3}\,dy = y - 2\sqrt{3}\,\arctan\left(\frac{1}{3}y\sqrt{3}\right)$$

Example 7

```
> Int((3*x-2)/(x^2-6*x+13),x)=int((3*x-2)/(x^2-6*x+13),x);
```

$$\int \frac{3x-2}{x^2-6x+13}\,dx = \frac{3}{2}\ln(x^2-6x+13) + \frac{7}{2}\arctan\left(\frac{1}{2}x-\frac{3}{2}\right)$$

```
>
```